
NAG Library Routine Document

F08FNF (ZHEEV)

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

F08FNF (ZHEEV) computes all the eigenvalues and, optionally, all the eigenvectors of a complex n by n
Hermitian matrix A.

2 Specification

SUBROUTINE F08FNF (JOBZ, UPLO, N, A, LDA, W, WORK, LWORK, RWORK, INFO)

INTEGER N, LDA, LWORK, INFO

REAL (KIND=nag_wp) W(N), RWORK(*)

COMPLEX (KIND=nag_wp) A(LDA,*), WORK(max(1,LWORK))

CHARACTER(1) JOBZ, UPLO

The routine may be called by its LAPACK name zheev.

3 Description

The Hermitian matrix A is first reduced to real tridiagonal form, using unitary similarity transformations,
and then the QR algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally)
the eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A,
Hammarling S, McKenney A and Sorensen D (1999) LAPACK Users’ Guide (3rd Edition) SIAM,
Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University Press,
Baltimore

5 Parameters

1: JOBZ – CHARACTER(1) Input

On entry: indicates whether eigenvectors are computed.

JOBZ ¼ N
Only eigenvalues are computed.

JOBZ ¼ V
Eigenvalues and eigenvectors are computed.

Constraint: JOBZ ¼ N or V .

2: UPLO – CHARACTER(1) Input

On entry: if UPLO ¼ U , the upper triangular part of A is stored.

If UPLO ¼ L , the lower triangular part of A is stored.

Constraint: UPLO ¼ U or L .

3: N – INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N � 0.

4: AðLDA,�Þ – COMPLEX (KIND=nag_wp) array Input/Output

Note: the second dimension of the array A must be at least max 1;Nð Þ.
On entry: the n by n Hermitian matrix A.

If UPLO ¼ U , the upper triangular part of A must be stored and the elements of the array
below the diagonal are not referenced.

If UPLO ¼ L , the lower triangular part of A must be stored and the elements of the array
above the diagonal are not referenced.

On exit: if JOBZ ¼ V , then A contains the orthonormal eigenvectors of the matrix A.

If JOBZ ¼ N , then on exit the lower triangle (if UPLO ¼ L ) or the upper triangle (if UPLO ¼ U )
of A, including the diagonal, is overwritten.

5: LDA – INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08FNF
(ZHEEV) is called.

Constraint: LDA � max 1;Nð Þ.

6: WðNÞ – REAL (KIND=nag_wp) array Output

On exit: the eigenvalues in ascending order.

7: WORKðmax 1;LWORKð ÞÞ – COMPLEX (KIND=nag_wp) array Workspace

On exit: if INFO ¼ 0, the real part of WORKð1Þ contains the minimum value of LWORK required
for optimal performance.

8: LWORK – INTEGER Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F08FNF
(ZHEEV) is called.

If LWORK ¼ �1, a workspace query is assumed; the routine only calculates the optimal size of the
WORK array, returns this value as the first entry of the WORK array, and no error message related
to LWORK is issued.

Suggested value: for optimal performance, LWORK � nbþ 1ð Þ � N, where nb is the optimal block
size for F08FSF (ZHETRD).

Constraint: LWORK � max 1; 2� Nð Þ.

9: RWORKð�Þ – REAL (KIND=nag_wp) array Workspace

Note: the dimension of the array RWORK must be at least max 1; 3� N� 2ð Þ.

10: INFO – INTEGER Output

On exit: INFO ¼ 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO ¼ �i, argument i had an illegal value. An explanatory message is output, and execution of
the program is terminated.

INFO > 0

If INFO ¼ i, the algorithm failed to converge; i off-diagonal elements of an intermediate tridiagonal
form did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix Aþ Eð Þ, where

Ek k2 ¼ O �ð Þ Ak k2,

and � is the machine precision. See Section 4.7 of Anderson et al. (1999) for further details.

8 Further Comments

Each eigenvector is normalized so that the element of largest absolute value is real and positive.

The total number of floating point operations is proportional to n3.

The real analogue of this routine is F08FAF (DSYEV).

9 Example

This example finds all the eigenvalues and eigenvectors of the Hermitian matrix

A ¼

1 2� i 3� i 4� i
2þ i 2 3� 2i 4� 2i
3þ i 3þ 2i 3 4� 3i
4þ i 4þ 2i 4þ 3i 4

0
BB@

1
CCA,

together with approximate error bounds for the computed eigenvalues and eigenvectors.

9.1 Program Text

Program f08fnfe

! F08FNF Example Program Text

! Mark 24 Release. NAG Copyright 2012.

! .. Use Statements ..
Use nag_library, Only: blas_zamax_val, ddisna, nag_wp, x02ajf, x04daf, &

zheev
! .. Implicit None Statement ..

Implicit None
! .. Parameters ..

Integer, Parameter :: nb = 64, nin = 5, nout = 6
! .. Local Scalars ..

Real (Kind=nag_wp) :: eerrbd, eps, r
Integer :: i, ifail, info, k, lda, lwork, n

! .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,:), work(:)
Complex (Kind=nag_wp) :: dummy(1)
Real (Kind=nag_wp), Allocatable :: rcondz(:), rwork(:), w(:), zerrbd(:)

! .. Intrinsic Procedures ..
Intrinsic :: abs, cmplx, conjg, max, nint, real

! .. Executable Statements ..
Write (nout,*) ’F08FNF Example Program Results’
Write (nout,*)

! Skip heading in data file
Read (nin,*)
Read (nin,*) n
lda = n
Allocate (a(lda,n),rcondz(n),rwork(3*n-2),w(n),zerrbd(n))

! Use routine workspace query to get optimal workspace.
! The NAG name equivalent of zheev is f08fnf

lwork = -1
Call zheev(’Vectors’,’Upper’,n,a,lda,w,dummy,lwork,rwork,info)

! Make sure that there is enough workspace for blocksize nb.
lwork = max((nb+1)*n,nint(real(dummy(1))))
Allocate (work(lwork))

! Read the upper triangular part of the matrix A from data file

Read (nin,*)(a(i,i:n),i=1,n)

! Solve the Hermitian eigenvalue problem
! The NAG name equivalent of zheev is f08fnf

Call zheev(’Vectors’,’Upper’,n,a,lda,w,work,lwork,rwork,info)

If (info==0) Then

! Print solution

Write (nout,*) ’Eigenvalues’
Write (nout,99999) w(1:n)

Write (nout,*)
Flush (nout)

! Normalize the eigenvectors so that the element of largest absolute
! value is real.

Do i = 1, n
Call blas_zamax_val(n,a(1,i),1,k,r)
a(1:n,i) = a(1:n,i)*(conjg(a(k,i))/cmplx(abs(a(k,i)),kind=nag_wp))

End Do

! ifail: behaviour on error exit
! =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft

ifail = 0
Call x04daf(’General’,’ ’,n,n,a,lda,’Eigenvectors’,ifail)

! Get the machine precision, EPS and compute the approximate
! error bound for the computed eigenvalues. Note that for
! the 2-norm, max( abs(W(i)) ) = norm(A), and since the
! eigenvalues are returned in descending order
! max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)))

eps = x02ajf()
eerrbd = eps*max(abs(w(1)),abs(w(n)))

! Call DDISNA (F08FLF) to estimate reciprocal condition
! numbers for the eigenvectors

Call ddisna(’Eigenvectors’,n,n,w,rcondz,info)

! Compute the error estimates for the eigenvectors

Do i = 1, n
zerrbd(i) = eerrbd/rcondz(i)

End Do

! Print the approximate error bounds for the eigenvalues
! and vectors

Write (nout,*)

Write (nout,*) ’Error estimate for the eigenvalues’
Write (nout,99998) eerrbd
Write (nout,*)
Write (nout,*) ’Error estimates for the eigenvectors’
Write (nout,99998) zerrbd(1:n)

Else
Write (nout,99997) ’Failure in ZHEEV. INFO =’, info

End If

99999 Format (3X,(8F8.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4)

End Program f08fnfe

9.2 Program Data

F08FNF Example Program Data

4 :Value of N

(1.0, 0.0) (2.0, -1.0) (3.0, -1.0) (4.0, -1.0)
(2.0, 0.0) (3.0, -2.0) (4.0, -2.0)

(3.0, 0.0) (4.0, -3.0)
(4.0, 0.0) :End of matrix A

9.3 Program Results

F08FNF Example Program Results

Eigenvalues
-4.2443 -0.6886 1.1412 13.7916

Eigenvectors
1 2 3 4

1 0.4836 0.6470 -0.4326 0.3809
0.0000 0.0000 0.1068 -0.0622

2 0.2912 -0.4984 -0.1590 0.4358
-0.3618 -0.1130 -0.5480 -0.0869

3 -0.3163 0.2949 0.5491 0.5241
-0.3696 0.3165 0.0000 0.0000

4 -0.4447 -0.2241 -0.2865 0.5719
0.3406 -0.2878 0.3037 0.2276

Error estimate for the eigenvalues
1.5E-15

Error estimates for the eigenvectors
4.3E-16 8.4E-16 8.4E-16 1.2E-16

___________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________

